;载入数据
NX=0l & NY=0l

OPENR,LUN,/GET_LUN,'.\COEFF_DAY.DAT',/F77_UNFORMATTED
READU,LUN,NX
READU,LUN,NY
X=fltARR(NX) & Y=fltARR(NY)
READU,LUN,X,Y
BDXX=fltARR(NX,NY) & BDYY=BDXX & BDXY=BDXX
READU,LUN,BDXX
READU,LUN,BDYY
READU,LUN,BDXY
FREE_LUN,LUN

BDYY=BDYY/REBIN(REFORM(Y^2,1,NY), NX, NY)
BDXY=BDXY/REBIN(REFORM(Y,1,NY), NX, NY)

ALPHA=X*!radeg
EK=(SQRT(Y*Y+1D0)-1D0)*0.511

toFile=1;设置1保存到文件否则输出到屏幕
N = 255;颜色层次数
;设备设置
device, decomposed=0;使用索引颜色系统
if toFile EQ 1 then begin
  set_plot, 'ps';保存到ps文件中
  device, filename='1.eps',$;文件名
          /color,$;开启颜色显示
          /encaps,$;使用eps
          /inches,$;单位为英寸
          xsize=6.0, ysize=10.0,$;图像大小
          /tt_font;开启truetype字体显示
endif else begin
  set_plot, 'win';设置窗口显示
  device, /tt_font;使用truetype字体
  window, 0, xsize=600, ysize=1000, title='Figure';打开新的窗口
endelse
loadct, 33;载入色表
tvlct, 0, 0, 0, 0;0号色设置为黑色


;格式控制
;边框
!X.omargin=[8,8]
!Y.omargin=[8,1]
!X.margin=[10,14]
!Y.margin=[0.5,0.5]
;
;边框位置，归一化坐标
;实际采用的margin控制，这里的坐标只用来控制了colorbar的位置
top=0.95
bottom=0.1
left=0.2
right=0.76
height=(top-bottom)/3.
gap=0.005;上下两幅图间的间隔
bargap=0.01
barwidth=0.03

;图片位置
fig1pos=[left,bottom+2.*height+gap,right,top]
fig2pos=[left,bottom+height+gap,right,bottom+2.*height-gap]
fig3pos=[left,bottom,right,bottom+height-gap]



;全局参数
!P.multi=[0,1,3,0,0];三行一列
!P.charthick=1.0;字体粗细
!P.charsize=1.5;字体大小
!P.font=1;字体种类
;!P.color=0;背景色
!X.charsize=2.5;x轴标记字体
!Y.charsize=2.5;y轴标记字体
!X.thick=2.0;x轴粗细
!Y.thick=2.0;y轴粗细

;x,z 标度数据
xr = [0,30,60,90]
zr = 10.0^(-8.0+5.0*findgen(N)/(N-1))
;作图使用的颜色表
colorindex=indgen(N)+1
Ntick=6;colorbar 上的标记数，与zr范围相对应
name=string(-8+indgen(Ntick),format="('!710!E',I0,'!N!7')");colorbar上的标记
;作图
contour,bdxx,alpha,ek,$;数据
        levels=zr,$;画线值
        /cell_fill,$;填充
        c_colors=colorindex,$;使用色表
        xstyle=1,xtickv=xr,$ ;x轴风格
        xtickformat='(a1)',xminor=3,$;不显示标签
        ystyle=1,/ylog,$ ;y轴风格
        ytitle='!8E!l!8k!N!7(MeV)';,$;y轴标题
        ;position=fig1pos;位置
xyouts,5,7,'!7(a)D!lxx!N!7',/data,charsize=2.0,charthick=2.0

contour,abs(bdxy),alpha,ek,$;数据
        levels=zr,$;画线值
        /cell_fill,$;填充
        c_colors=colorindex,$;使用色表
        xstyle=1,xtickv=xr,$ ;x轴风格
        xtickformat='(a1)',xminor=3,$;不显示标签
        ystyle=1,/ylog,$ ;y轴风格
        ytitle='!8E!l!8k!N!7(MeV)';,$;y轴标题
        ;position=fig2pos;位置
xyouts,5,7,'!7(b)D!lxy!N!7',/data,charsize=2.0,charthick=2.0

contour,bdyy,alpha,ek,$;数据
        levels=zr,$;画线值
        /cell_fill,$;填充
        c_colors=colorindex,$;使用色表
        xstyle=1,xticks=3,xtickv=xr,xminor=3,$ ;x轴风格
        xtitle='!Ma!7(Deg)',$;x轴标题
        ystyle=1,/ylog,$ ;y轴风格
        ytitle='!8E!l!8k!N!7(MeV)';,$;y轴标题
        ;position=fig3pos;位置
xyouts,5,7,'!7(c)D!lyy!N!7',/data,charsize=2.0,charthick=2.0
; colorbar 位置
barpos=[right+bargap,bottom+0.5*height,right+bargap+barwidth,bottom+2.5*height]
cgcolorbar,bottom=1,$
         /vertical,$;纵向颜色条
         /right,$;标签在颜色条右侧
         color=0,$;边框颜色
         divisions=Ntick-1,$;colorbar分块
         ticknames=name,$;标记字符串
         charsize=1.5,$
         title='!8D!l!8xx!N!8D!l!8xy!N!8D!l!8yy!N!7(s!E-1!N)',$;标题
         position=barpos;位置
if toFile EQ 1 then device, /close_file
end